Power u^:n _ _ _

Two cases occur: a numeric integer n, and a gerund n.

Numeric case . The verb u (or x&u) is applied n times. For example:

```
   (] ; +/\ ; +/\^:2 ; +/\^:0 1 2 3 _1 _2 _3 _4) 1 2 3 4 5
+---------+-----------+------------+-------------+
|1 2 3 4 5|1 3 6 10 15|1 4 10 20 35|1  2  3  4  5|
|         |           |            |1  3  6 10 15|
|         |           |            |1  4 10 20 35|
|         |           |            |1  5 15 35 70|
|         |           |            |1  1  1  1  1|
|         |           |            |1  0  0  0  0|
|         |           |            |1 _1  0  0  0|
|         |           |            |1 _2  1  0  0|
+---------+-----------+------------+-------------+
```

An infinite power n produces the limit of the application of u. For example, if x=:2 and y=:1, then x o.^:_ y is 0.73908, the solution of the equation y=Cos y. If n is negative, the obverse u^:_1 is applied |n times. The obverse (which is normally the inverse) is specified for six cases:

1. The self-inverse functions + - -. % %. |. |: /: [ ] C. p.

2. The pairs in the following lists:

```
<  <:  +.        +:  +~  *.        *:  *~  ^   ,:  ,~
>  >:  j./"1"_   -:  -:  r./"1"_   %:  %:  ^.  {.  (<.@-:@#)&{.

;:               #.  ".  ;~    3!:1  3!:3  p:    q:
;@(,&' '&.>"1)  #:  ":  >@{.  3!:2  3!:2  π(n)  */

\:      o.       j.     r.
/:@|.  %&(o.1)  %&0j1  %&0j1@^.
```

3. Obviously invertible bonded dyads such as -&3 and 10&^. and 1 0 2&|: and 3&|. and 1&o. and a.&i. as well as u@v and u&v if u and v are invertible.

4. Monads of the form v/\ and v/\. where v is one of + * - % = ~:

5. Obverses specified by the conjunction :.

Continued

Power `u^:n _ _ _`

Continued

6. The following cases merit special mention:

   `p:^:_1 n` gives the number of primes less than `n`, denoted by π(n) in math

   `q:^:_1` is `*/`

   `#^:_1` with a boolean left argument is “Expand” (whose fill atom `f` can be specified by fit, `b&#^:_1!.f`)

   `a&#.^:_1` produces the base-`a` representation

   `!^:_1` and `!&n^:_1` and `!&n&^:_1` produce the appropriate results

Gerund case. (Compare with the gerund case of the merge adverb `}`)

```
x u^:(v0`v1`v2)y   ↔   (x v0 y)u^:(x v1 y) (x v2 y)
x u^:(   v1`v2)y   ↔   x u^:([` v1`v2) y
  u^:(   v1`v2)y   ↔   u^:(v1 y) (v2 y)
```

Power u^:v _ _ _

The case of `^:` with a verb right argument is defined in terms of the noun right argument case (`u ^: n`) as follows:

```
x u ^: v y  ↔  x u^:(x v y) y
u ^: v y    ↔  u^:(v y) y
```

For example:

```
   x=: 1 3 3 1
   y=: 0 1 2 3 4 5 6
   x p. y
1 8 27 64 125 216 343

   x p. ^: (]>3:)"1 0 y
0 1 2 3 125 216 343

   a=: _3 _2 _1 0 1 2 3
   %: a
0j1.73205 0j1.41421 0j1 0 1 1.41421 1.73205

   * a
_1 _1 _1 0 1 1 1

   %: ^: * " 0 a
9 4 1 0 1 1.41421 1.73205

   *: a
9 4 1 0 1 4 9
```

The following monads are equivalent. (See the example of `^ T. _` in the definition of the Taylor Approximation `T.` .)

```
   g=: u ^: p ^: _
   h=: 3 : 't=. y. while. p t do. t=. u t end.'

   u=: -&3 [. p=: 0&<
   (g"0 ; h"0) i. 10
+------------------------+------------------------+
|0 _2 _1 0 _2 _1 0 _2 _1 0|0 _2 _1 0 _2 _1 0 _2 _1 0|
+------------------------+------------------------+
```